

Métodos Formales Industriales (MFI)

—prácticas—

Grado de Ingeniería en Informática

---

## Práctica 1: Programación en Maude

Santiago Escobar

---

Edificio DSIC 2 piso

## 1. Introducción

En esta práctica intentamos conseguir que el alumno refuerce en el laboratorio los conceptos introducidos sobre el lenguaje de especificación algebraico **Maude** durante las sesiones de teoría.

Los métodos formales en la Ingeniería Informática se definen sobre cuatro pilares básicos: (1) la existencia de un *modelo semántico* claro y conciso sobre el comportamiento de los sistemas software, (2) la existencia de una *representación* clara, detallada y sin ambigüedades que permita expresar sistemas software y asociarles una semántica concreta, (3) la existencia de un formalismo claro y detallado en el que se puedan definir *propiedades* de un sistema y en el que se pueda averiguar la validez o falsedad de dichas propiedades en función del modelo semántico y (4) la existencia de *técnicas eficientes y eficaces* para verificar dichas propiedades.

Los lenguajes de programación (o especificación) denominados declarativos son muy adecuados para ser utilizados durante el proceso de producción de software, especialmente en la definición de especificaciones prototípicas ejecutables, donde compiten con ventaja contra otro tipo de lenguajes de programación más complicados y con una semántica poco clara, como C#, Java, o Python.

## 2. El lenguaje de programación Maude

El lenguaje de programación **Maude** utiliza reglas de reescritura, como los lenguajes denominados funcionales tales como Haskell, ML, Scheme, o Lisp. En concreto, el lenguaje **Maude** está basado en la lógica de reescritura que permite definir multitud de modelos computacionales complejos tales como programación concurrente o programación orientada a objetos. Por ejemplo, **Maude** permite especificar objetos directamente en el lenguaje, siguiendo una aproximación declarativa a la programación orientada a objetos que no está disponible ni en lenguajes imperativos como C# o Java ni en lenguajes declarativos como Haskell.

El desarrollo del lenguaje **Maude** parte de una iniciativa internacional cuyo objetivo consiste en diseñar una plataforma común para la investigación, docencia y aplicación de los lenguajes declarativos. Se puede encontrar más información en:

<http://maude.cs.illinois.edu>

A continuación resumimos las principales características del lenguaje **Maude**. Sin embargo, hay un extenso manual y un “*primer*” (libro de introducción muy sencillo y basado en ejemplos) en la dirección web indicada antes. Existe también un libro sobre **Maude**, con ejemplares adquiridos por la Biblioteca General y de la Escuela, y accesible online en:

<https://link.springer.com/book/10.1007/978-3-540-71999-1>

### 2.1. Un programa en Maude

Un programa **Maude** está compuesto de diferentes módulos. Cada módulo se define entre las palabras reservadas **mod** y **endm**, si es un *módulo de sistema*, o entre **fmod** y **endfm**, si es un *módulo funcional*. Cada módulo incluye declaraciones de tipos y símbolos, junto con las reglas, encabezadas por **rl**, y las ecuaciones, encabezadas por **eq**, además de axiomas para listas o conjuntos, descritas con las palabras **assoc**, **comm**, e **id**. Las reglas describen transiciones entre distintas configuraciones (o estados) del modelo mientras que las ecuaciones y los axiomas describen cómo se ejecutan algunos de los símbolos, llamados *funciones*. En un módulo de sistema podremos incluir reglas y ecuaciones, pero en un módulo funcional sólo pueden aparecer ecuaciones.

Por ejemplo, podemos especificar el siguiente módulo funcional (sin reglas) que simula la función factorial:

```

fmod FACT is
  protecting NAT .

  op factorial : Nat -> Nat .

  var N : Nat .

  eq factorial(0) = 1 .
  eq factorial(s(N)) = factorial(N) * s(N) [otherwise] .
endfm

```

Maude es capaz de usar una doble representación para los números naturales importados del módulo NAT, como 0, 1, 2 y también como 0,  $s(0)$ ,  $s(s(0))$ . Estas dos ecuaciones son deterministas gracias al atributo “otherwise” y terminan para cada posible ejecución ya que van consumiendo el número natural. Una ejecución asociada a este programa del factorial es la siguiente (donde subrayamos el subtérmino reemplazado):

$$\begin{aligned}
& \underline{\text{factorial}(4)} \longrightarrow \underline{\text{factorial}(3)} * 4 \longrightarrow \underline{\text{factorial}(2)} * 3 * 4 \\
& \longrightarrow \underline{\text{factorial}(1)} * 2 * 3 * 4 \longrightarrow \underline{\text{factorial}(0)} * 1 * 2 * 3 * 4 \\
& \longrightarrow \underline{1} * 1 * 2 * 3 * 4 \longrightarrow 24
\end{aligned}$$

## 2.2. Errores comunes en los programas en Maude

- No finalizar las declaraciones con un espacio y un punto

```

sort Natural
op 0 : -> Natural.
op s : Natural -> Natural

```

- no dejar espacios entre los argumentos y un operador infijo (o mixfix), p.ej. “0+0” es un error y hay que escribir “0 + 0”
- no incluir paréntesis para desambiguar expresiones, p.ej. “p s s 0 + 0” es un error y hay que escribir “p s s (0 + 0)” ó “(p s s 0) + 0” en función de la sintaxis apropiada.

## 2.3. Tipos de datos predefinidos

Maude dispone de varios tipos de datos predefinidos incluidos en el fichero `prelude.maude` de la instalación. En concreto, se dispone del tipo `Bool` definido en el módulo `BOOL`, el tipo `Nat` en el módulo `NAT`, el tipo `Int` en el módulo `INT`, el tipo `Float` en el módulo `FLOAT`, y los tipos `Char` y `String` en el módulo `STRING`. Para hacer uso de alguno de esos tipos, sus operadores, deberemos importar el módulo donde se encuentran con una de las palabras reservadas<sup>1</sup> `including`, `protecting` o `extending`. Por ejemplo, el módulo `FACT` para factorial mostrado anteriormente importa el módulo `NAT` de los números naturales. El módulo `BOOL` es importado, por defecto, por todos los módulos cargados; esto se puede desactivar en el comando de llamada de Maude.

<sup>1</sup>El significado de estas acciones para importar módulos queda fuera del objetivo principal de esta práctica, pero se puede consultar en el manual. La palabra reservada `protecting` es la más segura y simple.

## 2.4. Declaración obligatoria de variables

Es obligatorio declarar el tipo de las variables antes de ser usadas en el programa, p.ej. la siguiente declaración de variables

```
vars N M : Nat .  
var NL : NatList .  
var NS : NatSet .
```

o añadirles el tipo directamente a las variables cuando vamos a usarlas, p.ej. “X:Nat + Y:Nat” .

## 2.5. Declaraciones de Tipos (Sorts), Símbolos Constructores y Variables

Una declaración de tipo tiene la forma:

```
sort T .
```

e introduce un nuevo tipo de datos T. Si queremos introducir varios tipos a la vez, escribiremos

```
sorts T1 ... Tn .
```

Después se definen los constructores que formarán los datos asociados a ese tipo de la forma:

```
op C : T1 T2 ... Tn -> T .
```

donde T<sub>1</sub>, T<sub>2</sub>, ..., T<sub>n</sub> son los tipos de los parámetros de ese símbolo. También podremos escribir

```
ops C1 ... Cn : T1 T2 ... Tn -> T .
```

y denota que todos los símbolos C<sub>1</sub>, ..., C<sub>n</sub> tienen el mismo tipo. Por ejemplo, las declaraciones de tipo:

```
sort Bool .  
ops true false : -> Bool .  
sort NatList .  
op nil : -> NatList .  
op _:_ : Nat NatList -> NatList .
```

introducen el tipo Bool con dos constantes true y false, y el tipo NatList (listas cuyos elementos son naturales, es decir de tipo Nat). Hay que tener en cuenta que Maude no soporta tipos de datos polimórficos, en el sentido de Haskell, por lo tanto no se pueden definir listas polimórficas, aunque sí se puede definir módulos paramétricos con estructuras de datos dentro, disponiendo de listas paramétricas. Sin embargo, es interesante fijarse en la forma de definir el operador clásico binario infijo de construcción de una lista, “\_:\_”, donde se indica que el primer argumento debe aparecer antes de los dos puntos mientras que el segundo detrás de los dos puntos. Una lista de enteros se podrá definir por lo tanto en notación infija como 0 : (1 : (2 : nil)) en vez de la notación prefija (0,:(1,:(2,nil))) simplemente indicando que el símbolo a utilizar es “\_:\_”. Esto es muy práctico y versátil ya que simplemente debemos indicar con un “\_” dónde va a aparecer el argumento, p.ej., se pueden definir símbolos tan versátiles como

```
op if_then_else-fi : Bool Exp Exp -> Exp .  
op for(_;_;_) { _ } : Nat Bool Bool Exp -> Exp .
```

En concreto en el ejemplo VENDING-MACHINE tenemos el símbolo

```
op _ : State State -> State
```

que denota que el carácter “vacío” es un símbolo válido usado para concatenar estados.

### 2.5.1. Tipos de datos ordenados y sobrecarga de operadores

En Maude se pueden crear tipos de datos ordenados o divididos en jerarquías. Por ejemplo, si en vez de importar los números naturales de Maude, definimos un nuevo tipo de datos para los números naturales usando el mismo sort `Nat`, podemos indicar que los números naturales se dividen en números naturales positivos, sort `NzNat`, y el cero, sort `Zero`, usando la palabra reservada `subsort` de la siguiente forma:

```
sorts Nat Zero NzNat .
subsort Zero < Nat .
subsort NzNat < Nat .
op 0 : -> Zero .
op s : Nat -> NzNat .
```

De esta forma, la expresión `s(0)` es de tipo `NzNat` y a la vez es de tipo `Nat`, mientras que no es de tipo `Zero`. E igualmente, la expresión `0` es de tipo `Zero` y `Nat`, pero no es de tipo `NzNat`.

Otra característica interesante del sistema de tipos de Maude es la sobrecarga de operadores. Por ejemplo, podemos reutilizar el símbolo `0` en el tipo de datos `Binary` sin ningún problema:

```
sorts Nat Zero NzNat .
subsort Zero < Nat .
subsort NzNat < Nat .
op 0 : -> Zero .
op s : Nat -> NzNat .

sort Binary .
op 0 : -> Binary .
op 1 : -> Binary .
```

En este caso, pueden surgir ambigüedades sobre algún término que se resuelven especificando el tipo detrás del término, por ejemplo `(0).Zero` ó `(0).Binary`. El sistema informará sólo de ambigüedades que no pueda resolver por su cuenta.

La unión de la sobrecarga de símbolos y los tipos ordenados le confiere una gran flexibilidad al lenguaje. Por ejemplo, podemos redefinir el anterior tipo de datos de lista de números naturales de la siguiente forma, donde `ENatList` denota lista vacía (es decir `nil`) y `NeNatList` denota lista no vacía de elementos:

```
sorts NatList ENatList NeNatList .
subsort ENatList < NatList .
op nil : -> ENatList .
subsort NeNatList < NatList .
op _:_ : Nat NeNatList -> NeNatList .
op _:_ : Nat ENatList -> NeNatList .
```

### 2.5.2. Propiedades avanzadas (o algebraicas) de los símbolos

El lenguaje Maude incorpora la posibilidad de especificar símbolos con propiedades algebraicas extra como asociatividad, conmutatividad, elemento neutro, etc. que facilitan mucho la creación de programas.

*Asociatividad*  $X + (Y + Z) = (X + Y) + Z$

*Conmutatividad*  $X + Y = Y + X$

*Elemento neutro*  $X + 0 = X$

De hecho, permiten definir estructuras de datos basadas en listas y conjuntos de forma muy sencilla e integrada dentro del lenguaje, mejorando el rendimiento de la ejecución y la compresión del programa o modelo. Por ejemplo, podemos redefinir el tipo de datos lista de números naturales, importando el módulo NAT, de la siguiente forma:

```
sorts NatList ENatList NeNatList .
subsort ENatList < NatList .
op nil : -> ENatList .
subsort Nat < NeNatList < NatList .
op _:_ : NatList NatList -> NeNatList [assoc] .
```

donde `_:_` es un símbolo asociativo, es decir no hacen falta los paréntesis para separar los términos. Nótese que los dos argumentos del símbolo `_:_` tienen que ser del mismo tipo para poder indicar que el símbolo es asociativo. Ahora Maude entiende que las siguientes expresiones significan exactamente lo mismo:

```
1 : 2 : nil
1 : (2 : nil)
(1 : 2) : nil
```

Otra posibilidad es añadir un elemento neutro al operador asociativo:

```
sorts NatList .
subsort Nat < NatList .
op nil : -> NatList .
op _:_ : NatList NatList -> NatList [assoc id: nil] .
```

donde ahora `_:_` es un símbolo asociativo y el término `nil` es el elemento neutro del tipo de datos, que por lo tanto se puede eliminar salvo cuando aparece solo. Ahora Maude entiende que las siguientes expresiones significan exactamente lo mismo:

```
1 : 2 : nil
1 : 2
nil : 1 : nil : 2 : nil
```

Y podemos añadir la propiedad de conmutatividad a la lista, creando el tipo de datos multiconjunto:

```
sorts NatMultiSet .
subsort Nat < NatMultiSet .
op nil : -> NatMultiSet .
op _:_ : NatMultiSet NatMultiSet -> NatMultiSet [assoc comm id: nil] .
```

donde la propiedad de conmutatividad indica que se puede intercambiar el orden de los elementos. Ahora Maude entiende que las siguientes expresiones significan exactamente lo mismo

```
0 : 1 : 2 : 1
0 : 1 : 1 : 2 : nil
1 : 0 : 2 : 1 : nil
nil : 1 : nil : 2 : nil : 1 : nil : 0 : nil
```

Y finalmente podemos añadir la propiedad de que no pueden haber elementos iguales, convirtiendo el multiconjunto en un conjunto:

```

sorts NatSet .
subsort Nat < NatSet .
op nil : -> NatSet .
op _:_ : NatSet NatSet -> NatSet [assoc comm id: nil] .
eq X:Nat : X:Nat = X:Nat .

```

donde la ecuación elimina aquellas ocurrencias repetidas de un término. Ahora Maude entiende que las siguientes expresiones significan exactamente lo mismo

```

0 : 1 : 2
0 : 1 : 1 : 1 : 1 : 2 : nil
1 : 0 : 2 : 1 : nil
nil : 1 : nil : 2 : nil : 1 : nil : 0 : nil

```

## 2.6. Declaración de Funciones

Aquellos operadores o símbolos que dispongan de reglas o ecuaciones que los definan son denominados *funciones* mientras que los que no dispongan de reglas o ecuaciones son denominados *constructores*. Nótese que es obligatorio en Maude declarar el tipo de todas las funciones, el tipo de todas las variables, etc. En otros lenguajes funcionales, como Haskell, esto no es necesario aunque se recomienda. En particular, esto puede ayudar a detectar fácilmente errores en el programa, cuando se definen funciones que no se ajustan al tipo declarado.

Respecto a las reglas/ecuaciones que definen las funciones, éstas pueden ser de la forma

```

rl f(t1,...,tn) => e .
eq f(t1,...,tn) = e .

```

donde  $t_1, \dots, t_n$  y  $e$  son términos. Las ecuaciones pueden etiquetarse con la palabra reservada **owise** (otherwise) y en ese caso se indica que sólo se aplicará si ninguna otra ecuación para ese símbolo es aplicable. **Cuidado** porque el **owise** sólo se puede aplicar a una ecuación, nunca a una regla ya que tienen un significado indeterminista. Por ejemplo, podemos escribir

```

fmod FACT is
  protecting NAT .

  op _! : Nat -> Nat .

  var N : Nat .

  eq 0 ! = 1 .
  eq N ! = (sd(N,1))! * N [owise] .

```

La expresión  $N - 1$  no es válida en Maude y hay que escribir  $sd(N, 1)$  que es la diferencia simétrica, que nunca devuelve un número negativo. Las funciones se pueden definir también mediante reglas/ecuaciones condicionales

```

crl f(t1,...,tn) => e if c .
ceq f(t1,...,tn) = e if c .

```

donde la condición  $c$  es un conjunto de emparejamientos de la forma  $t := t'$  separados por el operador  $\wedge$ . Un emparejamiento  $t := t'$  indica que el término  $t'$  debe tener la forma del término  $t$ , instanciando las variables de  $t$  si es necesario, ya que las variables de  $t$  pueden ser usadas en la

expresión `e` de la regla para extraer información de `t'`. Las ecuaciones condicionales sólo pueden aplicarse si la condición tiene éxito. También es posible definir una ecuación condicional en la que las guardas sean expresiones de tipo `Bool` en vez de `t := t'`; en ese caso se interpretan como `true := t`. Por ejemplo, podemos escribir la anterior función factorial de la siguiente forma (usando la resta simétrica `sd` para los números naturales):

```
ceq N ! = 1 if N == 0 .
ceq N ! = (sd(N,1))! * N if N /= 0 .
```

donde la igualdad `'=='` se evalúa a `true` si ambas expresiones son iguales y `'/='` se evalúa a `true` si ambas expresiones son distintas. Nótese que en este caso, se puede usar también el operador condicional `if_then_else_fi`:

```
eq N ! = if N == 0 then 1 else (sd(N,1))! * N fi .
```

Además, debido a los tipos de datos ordenados, se permiten expresiones lógicas de la forma `t :: T` que se evalúan a `true` si la expresión `t` es del tipo `T`. Por ejemplo, esto puede ser útil en ecuaciones condicionales como:

```
op emptyList : NatList -> Bool .
eq emptyList(NL) = NL :: ENatList .
```

donde se dice que una lista `NL` está vacía, es decir `emptyList` retorna `true`, si `NL` es del tipo `ENatList`.

### 2.6.1. Búsqueda eficiente de elementos en listas y conjuntos

Una ventaja de disponer de listas, multiconjuntos y conjuntos es que determinadas operaciones de búsqueda y emparejamiento de patrones resultan mucho más rápidas. De hecho, más rápidas que en otros lenguajes declarativos como Prolog o Haskell. Por ejemplo, la pertenencia de un elemento a una lista se realiza de forma secuencial en muchos lenguajes declarativos:

```
sort NatList .
op nil : -> NatList .
op _:_ : Nat NatList -> NatList .
```



```
op _in_ : Nat NatList -> Bool .
eq N:Nat in nil
  = false .
eq N:Nat in (X:Nat : XS:NatList)
  = (N:Nat == X:Nat) or-else (N:Nat in XS:NatList) .
```

Sin embargo, cuando disponemos de un operador asociativo con un elemento neutro, esta operación se hace de forma mucho más elegante y eficiente como sigue a continuación:

```
sort NatList .
subsort Nat < NatList .
op nil : -> NatList .
op _:_ : NatList NatList -> NatList [assoc id: nil] .

op _in_ : Nat NatList -> Bool .
eq N:Nat in (L1:NatList : N:Nat : L2:NatList)
```



```

= true .
eq N:Nat in L:NatList
= false [owise] .

```

La expresión “1 in 1 : 2 : 3” se puede ver como “1 in nil : 1 : 2 : 3” gracias a la propiedad del elemento neutro, donde “L1:NatList” se emparejaría con “nil”, “N:Nat” con “1” y “L2:NatList” con “2 : 3”; ocurre algo parecido con “2 in 1 : 2 : 3” y “3 in 1 : 2 : 3”. Esto tiene la ventaja de que es el propio sistema el que decide la mejor técnica de búsqueda y así no está restringido al mecanismo usado por el programador. Además, en el caso de un conjunto (o multiconjunto) es aún más simple gracias a la conmutatividad:

```

sort NatSet .
subsort Nat < NatSet .
op empty : -> NatSet .
op _:_ : NatSet NatSet -> NatSet [assoc comm id: nil] .

op _in_ : Nat NatSet -> Bool .
eq N:Nat in (N:Nat : L:NatSet)
= true .
eq N:Nat in L:NatSet
= false [owise] .

```

## 2.7. Entorno de Trabajo: el sistema Maude

El sistema Maude no dispone en la actualidad de un compilador a código máquina, aunque existen varias versiones en pruebas. Sólo se dispone de un intérprete que se carga con el comando:

```
$ maude
```

mostrando el siguiente texto de presentación:

```

\|||||/
--- Welcome to Maude ---
/|||||/
Maude 3.2.1 built: Feb 21 2022 18:24:38
Copyright 1997-2022 SRI International
Mon Feb 28 13:15:44 2022

```

Maude>

El sistema está disponible para diferentes versiones de UNIX (incluyendo Linux y Mac OS X). Existe, además, una versión de Maude para Windows descargable siguiendo el enlace “Maude for Windows” de la página <http://safe-tools.dsic.upv.es/maude/>. En los laboratorios docentes están instaladas las versiones de linux y Windows. Una vez arrancado el sistema, disponemos de los siguientes comandos:

`load < name > .` Lee y carga los distintos módulos almacenados en el archivo <name>. Los archivos de Maude se suelen crear con la extensión `.maude`.

`show modules .` Muestra los módulos cargados actualmente en el sistema.

`show module < module > .` Muestra el módulo <module> en pantalla.

`select < module > .` Selecciona un nuevo módulo para ser el módulo actual de ejecución de expresiones.

`reduce < expression > .` Evalúa la expresión `<expression>` con respecto al módulo actual usando sólo ecuaciones. También puede escribirse `“red < expression > .”`.

`reduce in < module > : < expresion > .` Evalúa la expresión `<expression>` con respecto al módulo `<module>`.

`rewrite < expression > .` Evalúa la expresión `<expression>` con respecto al módulo actual usando ecuaciones y reglas. También puede escribirse `“rew < expression > .”`.

`rewrite in < module > : < expresion > .` Evalúa la expresión `<expression>` con respecto al módulo `<module>`.

`search < expresion1 > => * < expression2 > .` Busca algún camino de ejecución que lleve del término `<expression1>` al término `<expression2>`. El término `<expression2>` puede tener variables mientras que `<expression1>` no. La expresión “=>\*” puede entenderse como buscar un camino de ejecución de `<expression1>` a `<expression2>` utilizando tantos pasos de ejecución como sean necesarios. También se puede sustituir por “=>!” indicando hasta que no sea posible dar más pasos de ejecución; o por “=>1” indicando un único paso de ejecución.

`search in < module > : < expresion1 > => * < expression2 > .` Busca algún camino de ejecución con respecto al módulo `<module>`.

`search [< limit >] < expresion1 > => * < expression2 > .` Busca varios caminos de ejecución, hasta un máximo de `<limit>`. Esto permite verificar propiedades de alcanzabilidad de forma positiva en sistemas con un número infinito de estados. **Cuidado** porque si no hay ninguna solución y el espacio de búsqueda es infinito el sistema no parará.

`cd < dir >` Permite cambiar de directorio (**cuidado** porque no lleva espacio ni punto al final).

`ls` Ejecuta el comando UNIX `ls` y muestra todos los ficheros en el directorio actual (**cuidado** porque no lleva espacio ni punto al final)

`quit` Salir del sistema (**cuidado** porque no lleva espacio ni punto al final).

La semántica operacional de *Maude* se basa en la lógica de reescritura y básicamente consiste en reescribir la expresión de entrada usando las reglas y ecuaciones del programa hasta que no haya más posibilidad. Para la ejecución de ecuaciones *Maude* utiliza una estrategia de ejecución *impaciente*, como ocurre en los lenguajes de programación imperativos como C o Pascal y en algunos lenguajes funcionales como ML, en vez de una estrategia de ejecución *perezosa*, como en el lenguaje funcional Haskell.

Por ejemplo, dada la función `_!` siguiente:

```
fmod FACT is
  protecting NAT .
  op _! : Nat -> Nat .
  var N : Nat .
  eq 0 ! = 1 . --- factorial de N=0 es 1
  eq N ! = (sd(N,1))! * N [owise] . --- factorial de N>0 es N*factorial de N-1
endfm
```

y asumiendo que está almacenada en el fichero `fact1.mau` escribiremos

```
$ mau
      \|||||/
      --- Welcome to Mau ---
      /|||||/
Mau 3.2.1 built: Feb 21 2022 18:24:38
Copyright 1997-2022 SRI International
Mon Feb 28 13:15:44 2022
Mau> load fact1.mau
Mau> red 4 ! .
reduce in FACT : 4 ! .
rewrites: 13 in 0ms cpu (0ms real) (481481 rewrites/second)
result NzNat: 24
Mau> quit
```

Y si especificamos una función con un comportamiento indeterminista, por ejemplo la función `_+-` que o bien incrementa o decrementa un número dado:

```
mod INC-DEC is
  protecting INT .
  op _+- : Int -> Int .
  var N : Int .
  rl N +- => N + 1 .
  rl N +- => N - 1 .
endm
```

y asumiendo que está almacenada en el fichero `masmenos.mau` escribiremos

```
$ mau
      \|||||/
      --- Welcome to Mau ---
      /|||||/
Mau 3.2.1 built: Feb 21 2022 18:24:38
Copyright 1997-2022 SRI International
Mon Feb 28 13:15:44 2022
Mau> load masmenos.mau
Mau> rew 1 +- .
rewrite in INC-DEC : 1 +- .
rewrites: 2 in 0ms cpu (0ms real) (71428 rewrites/second)
result NzNat: 2
Mau> quit
```

pero siempre va a devolver el valor 2 y si queremos buscar todas las soluciones usaremos el comando `search`:

```
$ mau
      \|||||/
      --- Welcome to Mau ---
      /|||||/
```

```
Maude 3.2.1 built: Feb 21 2022 18:24:38
Copyright 1997-2022 SRI International
Mon Feb 28 13:15:44 2022
```

```
Maude> load ejemplo3.mau
Maude> search 1 +- =>! X:Int .
search in INC-DEC : 1 +- =>! X:Int .
```

```
Solution 1 (state 1)
states: 3 rewrites: 4 in 0ms cpu (0ms real) (102564 rewrites/second)
X:Int --> 2
```

```
Solution 2 (state 2)
states: 3 rewrites: 4 in 0ms cpu (0ms real) (57971 rewrites/second)
X:Int --> 0
```

```
No more solutions.
states: 3 rewrites: 4 in 0ms cpu (0ms real) (44444 rewrites/second)
```

Es importante darse cuenta que en el módulo anterior no se puede ejecutar el comando `red 1 +-` porque está definido usando reglas y no ecuaciones

```
Maude> load masmenos.mau
Maude> red 1 +- .
reduce in INC-DEC : 1 +- .
rewrites: 0 in 0ms cpu (0ms real) (~ rewrites/second)
result Int: 1 +-
Maude> quit
```

### 3. Objetivo de la práctica

El objetivo de esta práctica consiste en responder a la serie de preguntas cortas que se muestran a continuación, modificando los programas de la forma requerida. Vamos a utilizar el proyecto QUIZIFY de PSW como inspiración para los ejercicios de toda la asignatura.

**Evaluación:** La evaluación de esta práctica se realizará con un examen sobre las diez preguntas incluidas a continuación.

#### 3.1. Lista de puntuaciones

El siguiente módulo funcional modela una lista de preguntas de un quiz o encuesta de QUIZIFY. Esa lista tendrá tantas posiciones como preguntas tenga el quiz. El símbolo `?` se añade a la lista para indicar que es una pregunta aún sin respuesta. Vamos a asumir solamente cuatro respuestas en los quiz o encuestas: A, B, C y D.

En esta sección creamos la lista usando el estilo de programación de Haskell. Una lista se representa con la estructura `cabeza @ cola`; noté que el símbolo `_@_` lo escoge el programador en Maude mientras que Haskell solo permite el símbolo `_::`. Por ejemplo, para el caso de una lista de 2 preguntas, las posibles combinaciones se representan como: `"A @ B @ nil"`, `"D @ ? @ nil"`, ó `"? @ ? @ nil"`. Sin embargo, vamos a añadir un identificador entre `< y >` a cada pregunta.

```
fmod QUIZIFY-HASKELL is
```

```

protecting NAT .

sorts Opt OptNat OptNatList .

ops A B C D ? : -> Opt [ctor] .
op _<_> : Opt Nat -> OptNat [ctor] .
op nil : -> OptNatList [ctor] .
op _@_ : OptNat OptNatList -> OptNatList [ctor] .

var L : OptNatList . vars N M : Nat .var O : Opt .

op add : OptNatList Nat -> OptNatList .
eq add(nil,N) = ? < N > @ nil .
eq add(O < M > @ L,N) = O < M > @ add(L,N) .

op first? : OptNatList ~> Nat .
eq first?(? < N > @ L) = N .
eq first?(A < N > @ L) = first?(L) .
eq first?(B < N > @ L) = first?(L) .
eq first?(C < N > @ L) = first?(L) .
eq first?(D < N > @ L) = first?(L) .
endfm

```

Nótese que como pueden haber listas donde todas las preguntas estén respondidas y la función `first?` se marca como `~> Nat` en vez de `-> Nat` para indicar que puede devolver un error y no retornar un número natural.

Se pueden lanzar las siguientes expresiones a ejecución:

```

reduce in QUIZIFY-HASKELL : add(A < 1 > @ B < 2 > @ nil, 3) .
result OptNatList: A < 1 > @ B < 2 > @ ? < 3 > @ nil

reduce in QUIZIFY-HASKELL : first?(A < 1 > @ ? < 2 > @ nil) .
result NzNat: 2

```

Basándonos en este módulo y usando tus propias palabras responde a las siguientes cuestiones:

(Pregunta 1) ¿Qué se obtiene tras ejecutar el siguiente comando y qué significa?

```
Maude> reduce in QUIZIFY-HASKELL : first?(A < 1 > @ B < 2 > @ nil) .
```

(Pregunta 2) ¿Cómo lo resolverías con una constante nueva, p.ej. “NoItem?”?

### 3.2. Lista de puntuaciones usando subtipos y axiomas en Maude

Maude permite definir una jerarquía de tipos para añadir más información a nuestros datos. Por ejemplo, usando la relación de subtipado (`subsort`), podemos definir el siguiente módulo funcional que nos permite distinguir entre `OptNatList` y `Opt?NatList` para saber cuando una lista tiene dentro el símbolo `?`. Además indicamos las listas usando la asociatividad y el elemento identidad `nil`.

```

fmod QUIZIFY-MAUDE is
  protecting NAT .

```

```

sorts Opt Opt? .
op ? : -> Opt? [ctor] .
ops A B C D : -> Opt [ctor] .
subsort Opt < Opt? .

sorts OptNat Opt?Nat .
subsort OptNat < Opt?Nat .
op _<_> : Opt Nat -> OptNat [prec 10] .
op _<_> : Opt? Nat -> Opt?Nat [prec 10] .

sorts OptNatList Opt?NatList .
subsort OptNatList < Opt?NatList .
op nil : -> OptNatList [ctor] .
subsort OptNat < OptNatList .
subsort Opt?Nat < Opt?NatList .
op _@_ : OptNatList OptNatList -> OptNatList [ctor assoc id: nil] .
op _@_ : Opt?NatList Opt?NatList -> Opt?NatList [ctor assoc id: nil] .

var L : OptNatList . vars N M : Nat . var L? : Opt?NatList .

op add : Opt?NatList Nat -> Opt?NatList .
eq add(L?,N) = L? @ ? < N > .


op first? : Opt?NatList ~> Nat .
eq first?(L @ ? < N > @ L?) = N .
endfm

```

En esta versión, podemos darnos cuenta que añadir un elemento es inmediato, mientras que antes había que recorrer toda la lista, y buscar el primer ? ahora se reduce a emparejamiento de términos gracias al uso de las variables L y L?. Esto no se puede hacer en Haskell.

Basándonos en este módulo y usando tus propias palabras responde a las siguientes cuestiones:

(Pregunta 3) Gráficamente, se puede representar las distintas conexiones de los tipos como un grafo:



Muestra un ejemplo de un término que sea de tipo Opt pero no Opt? y un término que sea de tipo OptNatList pero no OptNat ni Opt.

(Pregunta 4) ¿Para qué nos puede servir diferenciar el subtipo Opt dentro del tipo Opt??

(Pregunta 5) Para una lista de 2 preguntas, muestra algunos valores que pueden tomar una variable de tipo Opt?NatList y una variable de tipo OptNatList.

(Pregunta 6) ¿Qué tipo se obtiene tras ejecutar el siguiente comando y qué significa?

```
Maude> reduce in QUIZIFY-MAUDE : first?(nil @ A < 1 > @ nil @ B < 2 > @ nil) .
```

(Pregunta 7) ¿Podrías resolverlo de forma parecida a la pregunta 2?

### 3.3. Lista de puntuaciones mejorada usando indeterminismo en Maude

Partiendo del módulo QUIZIFY-MAUDE del ejercicio anterior se añade el siguiente módulo de reescritura que devuelve cualquier pregunta que no esté respondida.

```
mod QUIZIFY-CONCURRENT is
  protecting QUIZIFY-MAUDE .

  vars L1 L2 : OptNatList .
  vars L1? L2? : Opt?NatList .
  var N : Nat .

  op any? : Opt?NatList ~> Nat .
  rl any?(L1 @ ? < N > @ L2?) => N .
  rl any?(L1 @ ? < N > @ L2?) => any?(L1 @ L2?) .
endm
```

(Pregunta 8) Usando tus propias palabras, ¿qué se obtiene cuando se ejecuta el siguiente comando y por qué?

```
Maude> rew in QUIZIFY-CONCURRENT : any?(? < 1 > @ ? < 2 >) .
```

(Pregunta 9) ¿Y con el siguiente comando?

```
Maude> search in QUIZIFY-CONCURRENT : any?(? < 1 > @ D < 2 > @ ? < 3 >) =>* X:Nat .
```

(Pregunta 10) ¿Y con el siguiente comando?

```
Maude> search in QUIZIFY-CONCURRENT : any?(A < 1 > @ ? < 2 > @ ? < 3 >) =>* X:Nat .
```